
Sample Exit to force LE/CICS wide option settings LE/CICS 
wide option settings  
With this unsupported modification to the supplied CEECXITA 'LE/VSE Assembler User 
Exit for CICS' (skeleton CEEWCXIT) you can force specific run-time options to be active in 
a CICS-wide context. It's purpose is to illustrate a further technique of overriding specific run-
time options (aside of environment defaults via CEECOPT or application-specific settings via 
CEEUOPT).  
 
Please notice that it is required to allow the override (via the OVR-setting in CEECOPT.A) 
for each option you intend to customize. This sample actually refines the ENVAR run-time 
option via overriding environment variable LC_MONETARY.  
 
The modification has been successfully tested on VSE/ESA 2.7.0. 
No support or service of this sample is implied or acknowledged and use of it is 
completely at the user’s discretion and responsibility.  
 
This JCL will Assemble the sample user exit and then LINKEDT a new CEECXITA PHASE. 
It is recommended to save the default supplied LE/CICS environment wide assembler exit 
shipped with the product (before generating the modification version). Additional assist how 
to tailor JCL is provided in the sample attached. Please also be aware that the STACK option 
cannot be modified under CICS.  
 
Finally, it is required to recycle CICS and use the modified CEECXITA.PHASE at run-time 
(LIBDEF search chain).    
 
 
* $$ JOB JNM=CEEWCXIT,CLASS=0,DISP=D,LDEST=*,PDEST=* 
* $$ PUN DISP=I,PRI=9,CLASS=0 
// JOB CEEWCXIT 
/. C ***************************************************************** 
/. C                                                                 * 
/. C  PLEASE NOTE        IMPORTANT                                   * 
/. C --------------------------------------------------------------- * 
/. C                                                                 * 
/. C This is an unsupported modification to the supplied CEECXITA    * 
/. C 'LE/VSE Assembler User Exit for CICS' (skeleton CEEWCXIT).      * 
/. C                                                                 * 
/. C It shows an selective override of a default supplied LE/CICS-   * 
/. C wide option setting. In this case a different environment       * 
/. C variable has been chosen for the ENVAR run-time option.         * 
/. C                                                                 * 
/. C The modification has been successfully tested on VSE/ESA 2.7.0. * 
/. C No support or service of this sample is implied or acknowledged * 
/. C and use of it is completely at the users discretion and         * 
/. C responsibility.                                                 * 
/. C                                                                 * 
/. C This JCL will assemble the sample user exit and then LINKEDT    * 
/. C a new CEECXITA.PHASE member. As provided it is prepared as an   * 
/. C LE/CICS environment-wide assembler exit. In order to activate   * 
/. C it the new CEECXITA.PHASE must be the one used at run-time.     * 
/. C                                                                 * 
/. C You may wish to keep customized members in a user sublibrary    * 
/. C ahead of LE/VSE sublibraries. Therefore please tailor the JCL   * 
/. C before execution (and be aware that the STACK run-time option   * 
/. C cannot be modified under CICS).                                 * 
/. C                                                                 * 
/. C The sample JCL provided here catalogs the alternate exit        * 
/. C version to sublibrary PRD2.CONFIG and will not replace LE/VSE   * 
/. C provided members, supplied in PRD2.SCEEBASE.                    * 
/. C                                                                 * 
/. C ***************************************************************** 

/. C 
// PAUSE - Please ensure JCL has been tailored accordingly 
/. C Start of first run of IESINSRT 
/. C 
// ASSGN SYSIPT,SYSRDR 
// EXEC IESINSRT 
$ $$ LST DISP=D,CLASS=A,PRI=3 
// JOB CEECXIT 
* 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 
* Catalog the new exit module CEECXITA.OBJ to user library 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
* 
// LIBDEF *,SEARCH=PRD2.SCEEBASE 
// EXEC LIBR,PARM='MSHP' 
 ACC SUBLIB=PRD2.CONFIG 
 CATALOG CEECXITA.OBJ   REPLACE=YES 
* $$ END 
/. C 
/. C End of first run of IESINSRT 
/. C 
// OPTION DECK 
* 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 
* Assemble the new exit module CEECXITA.OBJ 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
* 
// LIBDEF *,SEARCH=PRD2.SCEEBASE 
// EXEC ASMA90,SIZE=512K,PARM='EXIT(LIBEXIT(EDECKXIT))' 
         TITLE 'LE/VSE Assembler User Exit under CICS' 
*/*******************************************************************/ 
*/*                                                                 */ 
*/*    Language Environment/VSE V1 R4 M3                            */ 
*/*                                                                 */ 
*/*    LICENSED MATERIALS - PROPERTY OF IBM                         */ 
*/*                                                                 */ 
*/*    5686-066-32-75K (C) COPYRIGHT IBM CORP. 1991,  2002          */ 
*/*    ALL RIGHTS RESERVED.                                         */ 
*/*                                                                 */ 
*/*   US Government Users Restricted Rights - Use, duplication or   */ 
*/*   disclosure restricted by GSA ADP Schedule Contract with IBM   */ 
*/*   Corp.                                                         */ 
*/*                                                                 */ 
*/*******************************************************************/ 
CEEBXITA CSECT 
CEEBXITA AMODE  ANY 
CEEBXITA RMODE  ANY 
         ENTRY  CEEBXITA 
*/*-----------------------------------------------------------------*/ 
*/* CEECXITA - Assembler User Exit under CICS                       */ 
*/*                                                                 */ 
*/* Function code:                                                  */ 
*/*    1 - first enclave initialization                             */ 
*/*    2 - first enclave termination                                */ 
*/*    3 - nested enclave initialization                            */ 
*/*    4 - nested enclave termination                               */ 
*/*    5 - PROCESS TERMINATION                                      */ 
*/*                                                                 */ 
*/* Register Usage:                                                 */ 
*/*    Input                                                        */ 
*/*       R1  - A(A(Assembler User Exit Plist cntrl blk)) - CEEAUE  */ 
*/*             NOTE: CEEAUE  contains the address of a 256 Byte    */ 
*/*                   work area initialized to zero which is used   */ 
*/*                   as the CEEBXITA's Save Area                   */ 
*/*       R15 - Entry Point Address                                 */ 
*/*       R14 - Return Address                                      */ 
*/*       R13 - Caller's save area address                          */ 
*/*       R12 - Pointer to CAA                                      */ 

© Copyright IBM Corp. 2005 



*/*    Output                                                       */ 
*/*       CEEAUE  altered                                           */ 
*/*    Work Regs                                                    */ 
*/*       R3 - Code Base                                            */ 
*/*       R2 - Base for User Exit Control Block (CEEAUE)            */ 
*/*                                                                 */ 
*/* History:                                                        */ 
*/*                                                                 */ 
*/*   10/26/98 (PVD)  PQ23382 - Performance enhancement. Sysroute   */ 
*/*                   of PQ14883. Default is return via immediate   */ 
*/*                   BR 14.                                        */ 
*/* *************************************************************** */ 
*/*    LE/VSE Version 1 Release 4 Modification Level 1 Changes :    */ 
*/*                                                                 */ 
*/* @01  GWH  Update for LE/VSE 1.4.1.                              */ 
*/*                                                                 */ 
*/* *************************************************************** */ 
*/*    LE/VSE Version 1 Release 4 Modification Level 2 Changes :    */ 
*/*                                                                 */ 
*/* $02  GWH  Update for LE/VSE 1.4.2.                         @02A */ 
*/*                                                                 */ 
*/* *************************************************************** */ 
*/*    LE/VSE Version 1 Release 4 Modification Level 3 Changes :    */ 
*/*                                                                 */ 
*/* $03  GWH  Update for LE/VSE 1.4.3.                         @03A */ 
*/*                                                                 */ 
*/*-----------------------------------------------------------------*/ 
         SPACE 2 
         USING CEEBXITA,R3        Code Base 
         USING CEEAUE,R2         User Exit Control Block 
         SPACE 1 
* 
*-------- IMPORTANT ------ IMPORTANT ---------- IMPORTANT ----------- 
*  The BR14 below is implemented as a default to reduce the LE 
*  default instruction path. Be sure to remove this when 
*  activating your CEECXITA program. 
* 
*        BR    14                 BRANCH BACK TO LE          @PQ23382 
*                                 IMMEDIATELY                @PQ23382 
*-------- IMPORTANT ------ IMPORTANT ---------- IMPORTANT ----------- 
* 
         STM   14,12,12(13)       Save Registers in caller's S.A. 
         LR    R3,R15             Set Module Base 
         B     AROUND 
         DC    CL8'CEECXITA'      Eye-catcher for Debugging 
         DC    CL8'V1.R4.M3'                                      @03C 
         DC    CL10'29/04/2002'                                   @03C 
* 
AROUND   DS    0H 
         L     R2,0(,R1)          Get the parm passed 
         L     R10,CEEAUE_A_WORK    Obtain the workarea 
*                                 Chain the Save Areas 
         ST    R10,8(,R13)        Old points to new 
         ST    R13,4(,R10)        New SaveArea points to old 
         LR    R13,R10            R13 points to new SaveArea 
* 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
*        BEGIN USER CODE BLOCK --- ADDED TO SKELETON 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
* 
         XR    R6,R6 
         LA    R6,A_CEEAUE_OPT_OVR  LOAD POINTER OPTION OVR STRING 
         ST    R6,CEEAUE_A_OPTIONS  STORE STRING TO CEEAUE_A_OPTIONS 
* 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 
*        END USER CODE BLOCK --- ADDED TO SKELETON 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 
* 

         L    R4,CEEAUE_FUNC        Get the function code. 
         C    R4,FRST_INIT_CODE   First EnclaveInitialization Exit? 
         BE   FRST_INIT_EXIT      Yes. Branch to it? 
* 
         C    R4,FRST_TERM_CODE   First Enclave Termination Exit? 
         BE   FRST_TERM_EXIT      Yes. Branch to it? 
* 
         C    R4,NSTD_INIT_CODE   Nested Enclave Initialization Exit? 
         BE   NSTD_INIT_EXIT      Yes. Branch to it? 
* 
         C    R4,NSTD_TERM_CODE   Nested Enclave Termination Exit? 
         BE   NSTD_TERM_EXIT      Yes. Branch to it? 
* 
         C    R4,PROC_TERM_CODE   Process Termination Exit? 
         BE   PROC_TERM_EXIT      Yes. Branch to it? 
* 
         B    RETURN              It is neither Init nor Term 
*                                 ...exit code. Simply Return 
         EJECT 
*/*=================================================================*/ 
*/*      First enclave initialization exit.                         */ 
*/*=================================================================*/ 
FRST_INIT_EXIT DS   0H 
         SPACE 1 
* 
*  To activate the abend codes table, uncomment the next 2 lines 
* 
*        LA    R4,ABEND_CODES     Get addr(abend codes table) 
*        ST    R4,CEEAUE_A_AB_CODES Store it in slot 
* 
         B     RETURN 
         SPACE 1 
*/*=================================================================*/ 
*/*      First enclave termination exit.                            */ 
*/*=================================================================*/ 
         SPACE 1 
* 
FRST_TERM_EXIT DS   0H 
         B     RETURN 
         SPACE 1 
*/*=================================================================*/ 
*/*      Nested enclave initialization exit.                        */ 
*/*=================================================================*/ 
         SPACE 1 
* 
NSTD_INIT_EXIT DS   0H 
         B     RETURN 
         SPACE 1 
*/*=================================================================*/ 
*/*      Nested enclave termination exit.                           */ 
*/*=================================================================*/ 
         SPACE 1 
* 
NSTD_TERM_EXIT DS   0H 
         B     RETURN 
         SPACE 1 
*/*=================================================================*/ 
*/*      Process termination exit.                                  */ 
*/*=================================================================*/ 
         SPACE 1 
* 
PROC_TERM_EXIT DS   0H 
*        B     RETURN 
* 
         SPACE 1 
*/*=================================================================*/ 
*/*               R E T U R N     T O    C A L L E R                */ 
*/*=================================================================*/ 

RETURN   DS    0H 
         L     R13,4(,R13)        Get A(caller's save area) 
         LM    R14,R12,12(R13)    Restore caller's regs 
         BR    R14                Go home 
         EJECT 
*              CONSTANTS AND WORKAREAS 
* 
FRST_INIT_CODE  DC  F'1'          First enclave initialization code 
FRST_TERM_CODE  DC  F'2'          First enclave termination code 
NSTD_INIT_CODE  DC  F'3'          Nested enclave initialization code 
NSTD_TERM_CODE  DC  F'4'          Nested enclave termination code 
PROC_TERM_CODE  DC  F'5'          Process termination code 
* 
* 
*   NOTE: The following table is not enabled (ignored) in CICS. 
* 
ABEND_CODES DS 0H 
* 
           DC  A(((ABEND_CODES_END-ABEND_CODES)/4)-1) 
* 
* 
ABEND_CODES_END DS 0H 
*------------------------------------------------------------------- 
* Condition token constants 
* 
*------------------------------------------------------------------- 
* Condition token constant for 'abend trapped'. 
* 
TOK_ABD_TRP DS 0CL8 
           DC  X'0004'            Severity 4 
           DC  X'0CB2'            Message number (3250) 
           DC  X'61'              Case 1 token 
*                                 Severity 4 condition 
           DC  C'CEE'             Facility ID 
* 
* 
* Condition token constant for 'operation exception' 
* 
TOK_OP_EXC DS 0CL8 
           DC  X'0003'            Severity 3 
           DC  X'0C81'            Message number (3201) 
           DC  X'59'              Case 1 token 
*                                 Severity 3 condition 
           DC  C'CEE'             Facility ID 
* 
* 
* Condition token constant for 'floating point divide exception' 
* 
TOK_FP_EXC DS 0CL8 
           DC  X'0003'            Severity 3 
           DC  X'0C8F'            Message number (3215) 
           DC  X'59'              Case 1 token 
*                                 Severity 3 condition 
           DC  C'CEE'             Facility ID 
* 
                     SPACE 1 
* 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
*          BEGIN SAMPLE DATA BLOCK --- ADDED TO SKELETON 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
* 
* Definition of Run-Time Option string set by this LE/CICS wide 
* Assembler User Exit 
* 
CEEAUE_OPT_OVR       DC  H'25' 
                     DC  C'ENVAR("LC_MONETARY=GERM")' 
A_CEEAUE_OPT_OVR     DC  A(CEEAUE_OPT_OVR) 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 

*          END SAMPLE DATA BLOCK --- ADDED TO SKELETON 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 
* 
*=================================================================== 
*         Parameter list passed to the User Exit. 
*         Addressed by Reg 1 = A(A(CEEAUE)) 
*=================================================================== 
CEEAUE               DSECT 0D 
CEEAUE_LEN           DS  F        Len of user exit Ctl Blk 
CEEAUE_FUNC          DS  F        Function Code 
*                                 1 ==> Initialization Exit 
*                                 2 ==> Termination Exit 
*                                 3 ==> Init Exit for nested enclave 
*                                 4 ==> Term Exit for nested enclave 
*                                 5 ==> Process termination exit 
CEEAUE_RETC          DS  F        Return or ABEND code 
CEEAUE_RSNC          DS  F        Return or ABEND Reason Code 
* 
CEEAUE_FLAGS         DS  0F       Flags 
CEEAUE_FLAG1         DS  X 
CEEAUE_ABTERM        EQU X'80'    0 = normal termination 
*                                 1 = abnormal termination. There 
*                                     was a condition of severity 2 or 
*                                     greater and the condition was 
*                                     not handled. 
CEEAUE_ABND          EQU X'40'    0 = terminate pgm with Return/Reason 
*                                 1 = terminate pgm with ABEND/Reason 
CEEAUE_DUMP          EQU X'20'    0 = Do not include a dump on an ABEND 
*                                 1 = Include a dump on an ABEND 
                     SPACE 1 
CEEAUE_FLAG2         DS  X        Reserved 
CEEAUE_FLAG3         DS  X        Reserved 
CEEAUE_FLAG4         DS  X        Reserved 
                     SPACE 1 
CEEAUE_A_CC_PLIST    DS  A        Ptr to Reg 1 upon invocation 
*                                 ...Initialization Exit Only 
CEEAUE_A_WORK        DS  A        Ptr to 256-byte work area 
CEEAUE_A_OPTIONS     DS  A        A(A(Run-Time Options string)) 
CEEAUE_USERWD        DS  F        User word. Retained thru Termination 
CEEAUE_A_AB_CODES    DS  A        NOTE: This field is ignored in CICS. 
CEEAUE_FBCODE        DS  A        Feedback code 
CEEAUE_PAGE          DS  F        Minimum value for page allocations 
* 
*   Condition token structure 
* 
TOK                  DSECT 0F 
TOK_DATA             DS CL8 
TOK_ISA              DS F 
* 
                     SPACE 1 
*/*=================================================================*/ 
*/*      E N D     O F     P A R A M E T E R    L I S T             */ 
*/*=================================================================*/ 
                     SPACE 1 
CEEBXITA CSECT 
R0       EQU   0 
R1       EQU   1 
R2       EQU   2 
R3       EQU   3 
R4       EQU   4 
R5       EQU   5 
R6       EQU   6 
R7       EQU   7 
R8       EQU   8 
R9       EQU   9 
R10      EQU   10 
R11      EQU   11 
R12      EQU   12 

R13      EQU   13 
R14      EQU   14 
R15      EQU   15 
         END 
/* 
/. C 
/. C Start of second run of IESINSRT 
/. C 
// EXEC IESINSRT 
/* 
* 
* <<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< 
* Linkedit CEECXITA.PHASE ... using new exit module CEECXITA.OBJ 
* 
* NOTE: You may wish to save shipped CEECXITA.PHASE (of sublibrary 
*       PRD2.SCEEBASE) now in case your JCL is tailored in a way to 
*       replace the IBM-supplied version. 
* >>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>>> 
* 
// LIBDEF *,SEARCH=(PRD2.CONFIG,PRD2.SCEEBASE) 
// PAUSE - Reminder for saving shipped CEECXITA.PHASE (if desired) 
// LIBDEF PHASE,CATALOG=PRD2.CONFIG 
// OPTION CATAL 
 INCLUDE CEE$CXIT 
/* 
// EXEC LNKEDT,PARM='MSHP' 
#& 
$ $$ EOJ 
* $$ END 
/. C 
/. C End of second run of IESINSRT 
/. C 
/& 
* $$ EOJ 
 
 
 
Trademarks 
 
The following terms are trademarks of International Business Machines Corporation in the 
United States, or other countries, or both:  
 

CICS, IBM, Language Environment, VSE/ESA, z/VSE 

 

Other company, product, or service names, may be the trademarks or service marks of others. 

 

Comments and Questions 
 

Comments or questions on this documentation are welcome. Please send your comments to: 

 

zvse@de.ibm.com 

